# DSolve returns unevaluated

I am trying to find a solution to this system of differential equations but the program gives the same output without any messages. I would like help. Please find the Mathematica code posted here:

s1 = I Exp[I κ x] (Derivative[1][y][x] - I/2 Δ y[x]) == f u[x] + f γ v[x] ;
s2 = I Exp[-I κ x] (Derivative[1][u][x] + I/2 δ u[x]) == f y[x];
s3 = I Exp[-I κ x] (Derivative[1][v][x] - I/2 δ v[x]) == f γ y[x] ;
system = {s1 , s2, s3};
s = DSolve[system, {y, u, v}, x]

• At first blush, this could indicate that DSolve does not know how to solve your equation. Do you have reason to believe that an analytical solution should exist? – MarcoB Mar 29 at 18:18
• Maple 2019 can solve. – Mariusz Iwaniuk Mar 29 at 18:23
• A system of two equations was resolved only after converting them into two differential equations of the second order so I think they can have a solution – Mohammed Omran Mar 29 at 18:25
• @Alrubaie What do you find confusing in this? OP included code already. Also, did you know that you can edit your comments within the first five minutes? See that Edit button? It would be helpful to keep the number of comments to a minimum, so important ones do not disappear. – MarcoB Mar 29 at 19:24

You need to help Mathematica a bit by removing the Exp factors.

Define your system of differential equations.

s1 = I Exp[I \[Kappa] x] (Derivative[1][y][x] - I/2 \[CapitalDelta] y[x]) == f u[x] + f \[Gamma] v[x];
s2 = I Exp[-I \[Kappa] x] (Derivative[1][u][x] + I/2 \[Delta] u[x]) == f y[x];
s3 = I Exp[-I \[Kappa] x] (Derivative[1][v][x] - I/2 \[Delta] v[x]) == f \[Gamma] y[x];
system = {s1, s2, s3}


Transform the system to cancel out the Exp factors.

u[x_] = E^(I x \[Kappa]) u1[x];
v[x_] = E^(I x \[Kappa]) v1[x];
y[x_] = y1[x];


Solve the transformed system.

DSolve[system, {y1, u1, v1}, x]


[verbose solution omitted]

Back-substitute to obtain the required solution.

• Thank you 'so much for this good idea it's really effective @Stephen Luttrell – Mohammed Omran Mar 29 at 20:16